

## Lesson 11: Laser Beam Shaper, all Spherical

The output from a laser has a non-uniform Gaussian intensity profile, and for some applications one would like to make it uniform. That is the job of a laser beam shaper.

### Approaches:


The job can be done in several ways. With simple lenses with spherical surfaces, it requires balancing rather large amounts of spherical aberration in a way that redistributes the light, reducing the energy density at the center of the beam while increasing it near the edges. It is easier to do with aspheric surfaces, where one has more control over the amount of aberration to be introduced, and even easier if one uses diffractive optical elements. The problem is that the latter are much more expensive than spherical lenses, so we want to see what we can do with those first.

With this in mind, we will try an all-spherical approach to start with, to determine how uniform we can make the beam and how many elements we will need. Let us start with a very simple setup, one that we will modify as we go along.

The problem is to convert a HeNe laser with a waist radius of 0.35 mm to a beam that is 10 mm in diameter and uniform to within 10%.

Here is the input file for our very primitive starting point:

```
RLE                ! Beginning of lens input file.
ID LASER BEAM SHAPER
WA1 .6328          ! Single wavelength
UNI MM            ! Lens is in millimeters
OBG .35 2         ! Gaussian object; waist radius -.35 mm; define full aperture as 2 * 1/e**2 point.
1 TH 22           ! Surface 2 is 22 mm from the waist.
2 RD -5 TH 2 GTB S ! Guess some reasonable lens parameters; use glass type SF6 from Schott catalog
SF6
3 UMC 0.3 YMT 5   ! Solve for the curvature of surface 3 so the marginal ray has an angle of 0.3;
                  ! find spacing so ray height is 5 mm on next surface
4 RD 20 TH 4 PIN 2 ! Guesses for surface 4
5 UMC 0 TH 50     ! Solve for curvature of 5 so beam is collimated.
7                ! Surfaces 6 and 7 exist
AFOCAL           ! because they are required for AFOCAL output.
END              ! End of lens input file.
```




(This system is in **AFOCAL** mode, which means that the output will be collimated. But transverse aberrations make no sense if the image is at infinity, so the program translates them into *angular* aberrations, no “perfect lens” required. But the program needs two dummy surfaces at the end, where it does the translation, here surfaces 6 and 7.)

First, let us examine how the energy density falls off from the center to the edge of the aperture. There are three ways to do this. The simplest is with the FLUX command:

**FLUX 100 P 3**

This input will show the flux on surface 3, which shows the expected falloff:



Another way uses the FLUX aberration. This form gives you more flexibility, since you can specify the aperture and field points yourself. Make a MACro with the following lines. (Note the symbol definition for DD):


```
DD: DO MACRO FOR AIP = -1 TO 1
COMPOSITE          ! Ready a composite definition.
CD1 P FLUX 0 0 AIP 0 3      ! Composite data number 1 is the flux at a relative Y- coordinate of AIP
                          ! (defined later) on surface 3.
= CD1
Z1 = FILE 1
= 1 + Z1
ORD = FILE 1
```

Run this MACro once, and then type STEPS = 100 and then DD. The program loops over values of AIP from -1 to 1 and plots the flux density. Let me explain the logic.

- CD1 P FLUX ... calculates the flux falloff at zone AIP (which is a loop variable) at surface 3.

- = CD1 is an equation, the result of which is automatically placed in FILE location 1.
- Z1 = FILE 1 picks up that value and puts it into variable Z1.
- = 1 + Z1 adds 1.0 to the result. This is then the total flux, since Z1 is the falloff.
- ORD = FILE 1 picks up this value and uses it for the ordinate of a plot, the abscissa of which is the loop variable, AIP.

Here is the plot:



Again, you see the Gaussian flux profile, evaluated out to twice the  $1/e^{**2}$  point according to the OBG definition. (The third way uses the **DPROP** diffraction propagation feature. That is more complicated to set up and run, but accounts for diffraction along the beam, which the other two do not.)


The object of this lesson is to make the flux as uniform as possible, with a target of 10% change over the aperture.

Here, we have simply guessed some starting lens dimensions, but already we see a hint of a solution: Notice how the marginal ray converges toward the axis while the center rays are more collimated. The energy will indeed be more

concentrated at the edges than before, which is a step in the right direction – but we also want the entire beam to be collimated, so we need a way to straighten out the rays afterwards. That is more difficult.

We suspect that we need more than just two elements, so let us add two more, using the WorkSheet. Click the

WorkSheet button , and click the Insert Element button in the WorkSheet toolbar, . Then click on the axis in the PAD display, to add an element to the right of surface 5. Do the same thing again, a little further to the right. Now the system should look like this:



We will try to optimize this system, but first let us make a Checkpoint, so we can instantly go back if things do not work

as we hope. Click the Checkpoint button, .

Now we need to set up an optimization MACro to see if we can even things out. Here is a start:

```


CHG
NOP
9 UMC
END

PANT                ! Start of variable parameter definitions.
VLIST RAD ALL      ! Vary all radii.
VLIST TH 3 5 6 7 8 ! Vary the airspaces and two thicknesses (so AEC works on those elements).
END


AANT                ! Start of merit function definition
AEC
ACC
LUL 100 1 1 A TOTL ! Prevent the system from growing too large; assign upper-limit of 150 on TOTL.
M 5 10 A P YA 0 0 1 0 9 ! Ask for a beam radius of 5 mm on surfaces 9 and 10
M 5 10 A P YA 0 0 1 0 10
M 0 1 A P FLUX 0 0 1 0 10 ! Ask for a flux falloff of zero at several zones on 10
M 0 1 A P FLUX 0 0 .98 0 10
M 0 1 A P FLUX 0 0 .97 0 10
M 0 1 A P FLUX 0 0 .96 0 10
M 0 1 A P FLUX 0 0 .95 0 10
M 0 1 A P FLUX 0 0 .94 0 10
M 0 1 A P FLUX 0 0 .93 0 10
M 0 1 A P FLUX 0 0 .92 0 10
M 0 1 A P FLUX 0 0 .91 0 10
M 0 1 A P FLUX 0 0 .85 0 10
M 0 1 A P FLUX 0 0 .8 0 10
M 0 1 A P FLUX 0 0 .7 0 10
M 0 1 A P FLUX 0 0 .5 0 10
M 0 1 A P FLUX 0 0 .3 0 10
GSO 0 1 5 P        ! Control the output ray OPD over an SFAN of 5 rays.
END                ! End of merit function definition.
SNAP
SYNO 100

```

Run this MACro, and the lens now looks like this (your results may differ, since the exact place you clicked to insert the elements is unpredictable.)




To see if things improved, we need to evaluate the flux uniformity again. Type `FLUX 100 P 10`. The flux is not improved.





It looks like we're not there yet. What to do?

Well, remember that we defined our Gaussian object so that "full aperture" is twice the  $1/e^{*2}$  point. That means we are trying to redistribute the energy so the very faint outer edge is as bright as the center. Maybe that's asking too much. Let's restore the Checkpoint again and edit the object specification. In the WorkSheet, enter the number 0 in the

surface box and click Update. This shows the current system specifications in the edit pane, including the object selection.




The object is currently **OBG 0.35 2.** Change this to **OBG .35 1** and click Update. Now, "full aperture" is at the  $1/e^{*2}$  point, not twice as far out as before. Save this as a new Checkpoint, run the last MACro again, and anneal for a few cycles.



Now the flux is somewhat better. We mentioned that it was not so easy to keep the ray angles under control at the same time one flattens the intensity profile. The OPDs are about .35 waves.

This seems to be about as good a balance as we can achieve with four elements. What happens if we add some more? Let's start with this design and add two more elements.



Now we need to add the new variables to the PANT file; also, since the merit function specifies the surface number where certain quantities are to be evaluated, and this number changes whenever we add an element, we can make things easier by changing the surface number to a special symbol, "LB1", which means "last but one". Since the highest surface in the lens is currently number 15, this symbol automatically turns into the number 14. Now, if we decide to add or delete elements as we search for a solution, we will not have to edit that number every time. We also add a GSR directive after the GSO, to better keep the ray angles under control, and we lower the weight on the OPD fan to better balance things.

```

CHG
NOP
13 UMC
END


PANT          ! Start of variable parameter definitions.
VLIST RAD ALL ! Vary all radii.
VLIST TH ALL
END

AANT          ! Start of merit function definition
AEC
ACC
LUL 100 1 1 A TOTL ! Prevent the system from growing too large
M 5 10 A P YA 0 0 1 0 LB1 ! Ask for a beam radius of 5 mm on surfaces 9 and 10
M 5 10 A P YA 0 0 1 0 LB2
M 0 1 A P FLUX 0 0 1 0 10 ! Ask for a flux falloff of zero at several zones
M 0 1 A P FLUX 0 0 .98 0 LB1
M 0 1 A P FLUX 0 0 .97 0 LB1
M 0 1 A P FLUX 0 0 .96 0 LB1
M 0 1 A P FLUX 0 0 .95 0 LB1
M 0 1 A P FLUX 0 0 .94 0 LB1
M 0 1 A P FLUX 0 0 .93 0 LB1
M 0 1 A P FLUX 0 0 .92 0 LB1
M 0 1 A P FLUX 0 0 .91 0 LB1
M 0 1 A P FLUX 0 0 .85 0 LB1
M 0 1 A P FLUX 0 0 .8 0 LB1
M 0 1 A P FLUX 0 0 .7 0 LB1
M 0 1 A P FLUX 0 0 .5 0 LB1
M 0 1 A P FLUX 0 0 .3 0 LB1
GSO 0 1 10 P          ! Control the output ray OPD over an SFAN of 5 rays.
GSR 0 50 10 P


END          ! End of merit function definition.
SNAP

```

We run this MACro, and then click the Anneal button , since we are getting close and want to be sure we find the optimum. There, we enter a starting temperature of about 50 and select the Quiet option to speed things up. After many iterations of annealing, our lens looks like this: (When you run these jobs yourself, the answers will usually differ, due to the random nature of the annealing process. But usually the quality will come out similar.)




And the flux is will within our target of 10% uniformity.



The OPD errors are now well under one wave. It appears that doing this job with all-spherical elements is possible but requires six elements.

Now we understand why people generally use aspherics or diffractive elements for this kind of job. Then there are fewer elements and better performance, so it can be worth the extra manufacturing trouble.

To finish this lesson, let us look at the ray pattern coming out of the system. Go to the footprint dialog (MFP), select surface 13, 10x scale, and ask for 600 rays. Here is the ray distribution:



This is pretty good for a beam shaper; rays are more spread out near the center and compressed near the edge, which is exactly the right way to make the beam more uniform.

Now we will use **DPROP** to evaluate the final envelope. In this design, the beam is expanded early on, from surface 3, and diffraction will not play a significant role thereafter. So we can run DPROP and the calculation will be purely geometric.


Here is the input we need to run DPROP:

```

CHG
CFIX
1 TH 0
END
DPROP P 0 0 13 SURF 3 R RESAMPLE

```

Here, we changed all apertures to fixed values, which is always recommended when running DPROP, since that program will often examine a larger area than the lenses allow, if diffraction can send even a small amount of energy there. The fixed apertures prevent this. The wavefront on surface 13 now looks like this:



To accurately model a Gaussian beam with real rays, one should expand the beam early on; then diffractive propagation is no longer an issue. In the next lesson we will develop a beam shaper with aspheric surfaces to illustrate the effect of those very useful variables.

This lesson was intended to show you some of the tools you have available and illustrate what to look out for and how to deal with the inevitable surprises and tradeoffs that turn up when you face a difficult and unfamiliar challenge.